
Theory and Computation in the Search for Special Surfaces

Let K be an algebraically closed field, let Pn be the n-dimensional projective space over
K, and let V and W be two disjoint irreducible projective varieties in Pn. We denote by
J(V, W ) the union of the lines in Pn joining V to W . Then J(V, W ) is a projective variety.
This variety is called the join of V and W .

Exercise 1. (a) Write a Macaulay 2 script for computing the ideal of the join of two given
varieties. Let L1 and L2 be two skew lines in P4. Compute the ideal of J(L1, L2).
(b) Make a Macaulay 2 function for computing the ideal of the secant variety to a given
variety (use the script you wrote in Exercise 1). Use your script to compute the ideal of
the secant variety to the Veronese variety in P5.

Hint. Let A = [a0 : · · · : an] and B = [b0 : · · · : bn] be points of V and W respectively. Then
any point R = [z0 : · · · : zn] of the line passing through A and B is given by z0

...
zn

 = s

 a0
...

an

 + t

 b0
...
bn

(1)

for some [s : t] ∈ P1. Let {f1, . . . , fl} and {g1, . . . , gm} be the generating sets for I(V ) and
I(W ) respectively. Then the ideal defining J(V, W ) is obtained by solving the system of
equations f1 = · · · = fl = 0, g1 = · · · = gm = 0 and (1) for z0, . . . , zn−1 and zn. By replacing
xi’s by ai’ and bi’s, we obtained new ideals I(V )a in K[a0, . . . , an] and I(W )b in K[b0, . . . , bn]
respectively. Let I be the ideal in the following new ring:

K[a0, . . . , an, b0, . . . , bn, s, t, z0, . . . , zn]

generated by the equations in (1), and let J = I + I(V )a + I(W )b. Saturate J with respect
to the ideal (s, t):

J ′ = J : (s, t)∞,

because we do not want s and t to vanish at the same time (recall that [s : t] ∈ P1). Take
the intersection J̄ = J ′ ∩K[z0, . . . , zn]. Then V (J̄) = J(V, W ).

Exercise 2. Construct a smooth rational surface X in P4, which is isomorphic to P2 blown
up in eight points embedded by the linear system of the following type:∣∣∣∣∣4L− 2E0 −

7∑
i=1

Ei

∣∣∣∣∣ ,

where L is the pullback from P2 a line, while the Ei are the exceptional curves of the blowup.
Compute the degree and sectional genus of X.

Exercise 3. Construct the union P of four planes P0, P1, P2 and P3 in P4 such that

(a) P0 intersects each of P1, P2 and P3 in a line;
(b) For 1 ≤ i < j ≤ 3, Pi and Pj meet at a single point.

Check that the ideal of U is generated by 7 cubics. Take two general cubics to get the ideal
of the surface X linked to U in a (3, 3) complete intersection. What is this surface? Any
line in Pi, i ∈ {1, 2, 3}, is a 3-secant line to X.

Theorem 1 (Beilinson, 1978). For any sheaf F on Pn, there is a complex K· with

Ki =
⊕

H i−j(Pn,F ⊗OPn(j))⊗ Ω−j(−j)

such that

H i(K·) =

{
F if i = 0,

0 otherwise.

Exercise 4. Let E be a locally free sheaf on P3 whose cohomology table for hi(P3, E(j)) in
the range −3 ≤ j ≤ 0 of twists is of the following type:

//
j

OO
i

2

2 2
hiE(j)

−3 −2 −1 0

0

1

2

3

Use Beilinson’s theorem to determine the type of the monad for E . Then construct the
differentials of the monad. How do you check that your sheaf is locally free?

Exercise 5. Let X be an elliptic conic bundle in P4. Compute the image of the adjunction
map Φ|H+K| : X → P2. (Use the script for the ideal of X).

You can find more exercises in the last two pages!

KK=ZZ/32003

-------------------------------

-- Example: Veronese surface --

-------------------------------

ringP2=KK[x_0..x_2];

ringP4=KK[y_0..y_4];

veronese=basis(2,ringP2)*random(ringP2^6,ringP2^5);

I=ker map(ringP2,ringP4,veronese);

hilbertPolynomial I

--------------------------------------

-- Example: Veronese surface in P^4 --

--------------------------------------

ringP4=KK[y_0..y_4];

veronese=basis(2,ringP2)*random(ringP2^6,ringP2^5);

I=ker map(ringP2,ringP4,veronese);

hilbertPolynomial I

-- Is V(I) smooth?

singI=minors(codim I,jacobian I)+I;

codim singI

betti I

--------------------------------

-- Example: del Pezzo surface --

--------------------------------

randomPoints=(N)->(

i:=1;

idl:=ideal random(ringP2^{0},ringP2^{2:-1});

while i<N do (

idl=intersect(idl,ideal random(ringP2^{0},ringP2^{2:-1}));

i=i+1;

);

idl)

points=randomPoints(5);

hilbertPolynomial points

del=gens points*map(source gens points,basis(3,points));

delPezzo=trim ker map(ringP2,ringP4,del);

betti res delPezzo

------------------------------------------------------------------

-- Example: Veronese surface in P^4 and elliptic quintic scroll --

------------------------------------------------------------------

veronese=basis(2,ringP2)*random(ringP2^6,ringP2^5);

I=ker map(ringP2,ringP4,veronese);

betti I

V=ideal(gens I*random(source gens I,ringP4^{2:-3}));

J=V:I;

betti J

X=Proj(ringP4/J);

HH^1(OO_X)

------------------------------------

--Example: nullcorellation bundle --

------------------------------------

sortedBasis=(i,E)->(

m:=basis(i,E);

p:=sortColumns (m, MonomialOrder=>Descending);

m_p);

beilinson1=(e,dege,i,S)->(

E:=ring e;

mi:=if i<0 or i>=numgens E then map(E^1,E^0,0)

else if i === 0 then id_(E^1)

else sortedBasis(i+1,E);

r:=i-dege;

mr:= if r<0 or r>=numgens E then map(E^1,E^0,0)

else sortedBasis (r+1,E);

s:=numgens source mr;

if i===0 and r===0 then substitute(map(E^1,E^1,{{e}}),S)

else if i>0 and r===i then substitute(e*id_(E^s),S)

else if i>0 and r===0 then

(vars S)*substitute(contract(diff(e,mi),transpose mr),S)

else substitute(contract(diff(e,mi),transpose mr),S));

U=(i,S)->(

if i<0 or i>=numgens S then S^0

else if i===0 then S^1

else coker koszul(i+2,vars S)**S^{i});

beilinson=(o,S)->(

coldegs:=degrees source o;

rowdegs:=degrees target o;

mats:=table(numgens target o, numgens source o,

(r,c)->(

rdeg=first rowdegs#r;

cdeg=first coldegs#c;

overS=beilinson1 (o_(r,c),cdeg-rdeg,cdeg,S);

map(U(rdeg,S),U(cdeg,S),overS)));

if #mats===0 then matrix(S,{{}})

else matrix(mats));

ringP3=KK[x_0..x_3];

E=KK[e_0..e_3,SkewCommutative=>true];

-- indecomposable

alpha=map(E^{-1},E^{-3},{{e_0*e_2+e_1*e_3}});

alpha=beilinson(alpha,ringP3);

betti alpha

omega1=U(1,ringP3)

F=prune coker(presentation omega1|map(target presentation omega1,,alpha));

fF=res F;

betti fF

rank F

Ft=syz transpose fF.dd_1;

-- Check F is locally free.

codim minors(2,Ft)

codim minors(3,Ft)

-- decomposable

alpha=map(E^{-1},E^{-3},{{e_0*e_2}});

alpha=beilinson(alpha,ringP3);

betti alpha

omega1=U(1,ringP3)

F=prune coker(presentation omega1|map(target presentation omega1,,alpha));

fF=res F;

betti fF

rank F

Ft=syz transpose fF.dd_1;

codim minors(2,Ft)

codim minors(3,Ft)

------------------------------------

-- Example: elliptic conic bundle --

------------------------------------

ringP4=KK[x_0..x_4];

E=KK[e_0..e_4,SkewCommutative=>true];

beta=random(E^{1:0},E^{1:-1,1:-2})

betti syz beta

alpha=(syz beta)*random(source syz beta,E^{-3});

beta=beilinson(beta,ringP4);

alpha=beilinson(alpha,ringP4);

F=prune homology(beta,alpha);

fF=res F;

betti fF

phi=presentation F|random(target presentation F,ringP4^{4:-1});

fphit=res (prune coker transpose phi,LengthLimit=>2);

betti fphit

I=ideal fphit.dd_2;

hilbertPolynomial I

------------------------------

-- Example: Bordiga surface --

------------------------------

restart

kk=ZZ/32003

S=kk[a..e]

I=ideal(

490*a^3+12221*a^2*b+1663*a*b^2-14146*b^3-6390*a^2*c-8829*a*b*c

+6322*b^2*c-9839*a*c^2-7186*b*c^2-4124*c^3-6141*a^2*d+15804*a*b*d

-10963*b^2*d-14244*a*c*d+3292*b*c*d+12528*c^2*d+1470*a*d^2

+9748*b*d^2-4690*c*d^2-11431*d^3-10463*a^2*e+965*a*b*e

+15018*b^2*e+3510*a*c*e-13207*b*c*e-12332*c^2*e+5496*a*d*e

-1110*b*d*e-14202*c*d*e+4737*d^2*e+11286*a*e^2+3759*b*e^2

-12124*c*e^2-8044*d*e^2-8400*e^3,

-4212*a^3-6354*a^2*b+8359*a*b^2-15620*b^3-7605*a^2*c+14070*a*b*c

+1533*b^2*c-6538*a*c^2+1929*b*c^2+6483*c^3+2503*a^2*d

-13769*a*b*d-10716*b^2*d+15442*a*c*d-10241*b*c*d+4046*c^2*d

-7595*a*d^2-4553*b*d^2-5535*c*d^2-12507*d^3-8300*a^2*e

+9255*a*b*e-166*b^2*e-3948*a*c*e-887*b*c*e+12166*c^2*e

-12019*a*d*e-4986*b*d*e-9965*c*d*e-2795*d^2*e-11300*a*e^2

-9200*b*e^2+2201*c*e^2+771*d*e^2-10878*e^3,

-13422*a^3-13098*a^2*b+165*a*b^2-15751*b^3-9062*a^2*c

+3931*a*b*c+10465*b^2*c+3947*a*c^2+15856*b*c^2-13765*c^3

+12661*a^2*d-759*a*b*d+2367*b^2*d-5578*a*c*d-10985*b*c*d

+987*c^2*d-9772*a*d^2-1366*b*d^2+2179*c*d^2-13964*d^3

-11350*a^2*e-1598*a*b*e+2754*b^2*e-9928*a*c*e+13439*b*c*e

+15037*c^2*e-9065*a*d*e+12521*b*d*e+9348*c*d*e-10519*d^2*e

-13803*a*e^2+6223*b*e^2-6660*c*e^2+9736*d*e^2+5263*e^3,

14635*a^3+2059*a^2*b-5482*a*b^2-14747*b^3+1035*a^2*c-4988*a*b*c

+11895*b^2*c+13195*a*c^2+7097*b*c^2-8995*c^3+1144*a^2*d

-11329*a*b*d-14481*b^2*d-4991*a*c*d+14404*b*c*d+10242*c^2*d

-6921*a*d^2-6661*b*d^2-15585*c*d^2+9999*d^3+4902*a^2*e

-12029*a*b*e-681*b^2*e-2020*a*c*e+14081*b*c*e+15876*c^2*e

+7994*a*d*e+7138*b*d*e-3671*c*d*e+11424*d^2*e+380*a*e^2

-11858*b*e^2+9754*c*e^2+13383*d*e^2+6735*e^3);

betti I

fI=res I;

betti fI

ringP2=kk[s_0..s_2];

ringP4xP2=kk[a..e,s_0..s_2,MonomialOrder=>Eliminate 5];

omegaX=coker transpose fI.dd_2;

phi=sub(vars ringP2, ringP4xP2)*sub(presentation omegaX,ringP4xP2);

adjunction=ideal(oo);

adjunctionsat=saturate(adjunction,ideal(a..e));

J=ideal sub(selectInSubring(1,gens gb oo),ringP2);

codim J

psi=contract(sub(vars S,ringP4xP2),transpose phi)

points=sub(minors(4,psi),ringP2);

hilbertPolynomial points

betti points

Lemma 1 (Terracini’s lemma). Let X be a smooth variety in Pn and let p1 · · · , ps be s
generic points on X. Then for a generic p ∈ 〈p1, · · · , ps〉,

dim σs(X) = dim〈Tp1X, . . . , TpsX〉,
where Tpi

X is the projectivized tangent space to X in Pn at pi.
Let X be a smooth variety in Pn and let p1 and p2 be two generic points on X. Then for

a generic p ∈ 〈p1, p2〉,
dim Sec(X) = dim〈Tp1X,Tp2X〉,

where Tpi
X is the projectivized tangent space to X in Pn at pi.

Exercise 6. Let X = G(3, 6) be the Grassmaniann of 3-planes in a 6-dimensional vector
space. Use Terracini’s lemma to prove that Sec(X) has the expected dimension.

Hint. Let V be a 6-dimensional vector space with basis e0, . . . , e5 and let p1 and p2 be the
points of X corresponding to e0 ∧ e1 ∧ e2 and e3 ∧ e4 ∧ e5. Try to prove that

dim〈Tp1X, Tp2X〉 = 19.

You can use the following fact: Let us denote by W the following subspace of
∧3 V :

V ∧ e1 ∧ e2 + e0 ∧ V ∧ e2 + e0 ∧ e1 ∧ V.

Note that Tp1X = P(W ). Let W⊥ =
{
w ∈

∧3 V |w ∧ e0 ∧ e1 ∧ e2

}
. The orthogonal com-

plement W⊥ can be identified with [(e0, e1, e2)
2]3, where (e0, e1, e2) is the ideal of the exte-

rior algebra
⊕6

i=0

∧i V generated by e0, e1 and e2 and [(e0, e1, e2)
2]3 is the degree 3 part of

(e0, e1, e2)
2.

Exercise 7. Let s ∈ H0(P3, E(1)) be a general section of E(1). Compute the ideal of the
zero locus Z of s. Answer the following questions:

(a) What is the dimension of Z? How about the degree and arithmetic genus of Z?
(b) Is Z smooth?
(c) How many components does Z have?

Exercise 8. Let M be the family of vector bundles whose cohomology tables are as given
in Exercise 4. Compute the dimension of the Zariski tangent space to M at the point
corresponding to the example you constructed in Exercise 4.


